unittest
{
//	assert( [ "5 3" ].parse.expand.solve == "5\n4 3\n1 2\n3 1\n4 5\n2 3" );
	assert( [ "5 8" ].parse.expand.solve == "-1" );
}



import std.algorithm;
import std.conv;
import std.range;
import std.stdio;
import std.typecons;

void main()
{
	stdin.byLineCopy.parse.expand.solve.writeln;
}

auto parse( Range )( Range input )
if( isInputRange!Range && is( ElementType!Range == string ) )
{
	auto nk = input.front.split.to!( long[] );
	return tuple( nk[ 0 ], nk[ 1 ] );
}

auto solve( long n, long k )
{
	// 最短距離が 2 であるような頂点対の個数の上限を計算
	auto m = ( n - 1 ) * ( n - 2 ) / 2;
	if( m < k ) return "-1";
	
	// うにグラフを作成（頂点対の個数が上限になる）
	auto anss = new long[ 2 ][ n - 1 ];
	foreach( i, ref ans; anss )
	{
		ans = [ i + 1, n ];
	}
	
	// うにグラフの外側の頂点同士を辺で結ぶ（頂点対の個数が1個減る）
	for( auto f = 1L; f < n - 1 && k < m; f++ )
	{
		for( auto t = f + 1; t < n && k < m; t++ )
		{
			anss ~= [ f, t ];
			m--;
		}
	}
	
	return anss.length.to!string ~ "\n" ~ anss.map!( ( ans ) => ans.to!( string[] ).join( " " ) ).join( "\n" );
}
